#!/usr/bin/env python
from misc.utility.scons_hints import *

Import("env")

# Thirdparty code

thirdparty_obj = []

thirdparty_dir = "#thirdparty/misc/"
thirdparty_sources = [
    "mikktspace.c",
    "qoa.c",
]

thirdparty_sources = [thirdparty_dir + file for file in thirdparty_sources]

env_thirdparty = env.Clone()
env_thirdparty.disable_warnings()
env_thirdparty.add_source_files(thirdparty_obj, thirdparty_sources)
env.scene_sources += thirdparty_obj

# Redot source files

scene_obj = []

env.add_source_files(scene_obj, "*.cpp")
env.scene_sources += scene_obj

# Needed to force rebuilding the scene files when the thirdparty code is updated.
env.Depends(scene_obj, thirdparty_obj)

SConscript("2d/SCsub")
if not env["disable_3d"]:
    SConscript("3d/SCsub")
